Ricerca Operativa - Laboratorio

Lezione 5 - Modelli di Programmazione Lineare Intera

Docente: Luigi De Giovanni

Dipartimento di Matematica "Tullio Levi-Civita"
Università degli Studi di Padova

luigi@math.unipd.it
https://www.math.unipd.it/~luigi/

Corso di Laurea Magistrale in Matematica
Università degli Studi di Padova
a.a. 2019–2020

Programmazione Lineare Intera e Lineare Mista Intera

Programmazione Lineare Intera (PLI - *ILP*):

- Funzioni del problema lineari
- Le variabili sono vincolate ad assumere valori interi

Programmazione Lineare Intera e Lineare Mista Intera

Programmazione Lineare Intera (PLI - *ILP*):

- Funzioni del problema lineari
- Le variabili sono vincolate ad assumere valori interi

Programmazione Lineare Intera Mista (PLIM - *MILP*):

- Funzioni del problema lineari
- Alcune variabili sono vincolate ad assumere valori interi, altre sono continue

Interezza delle variabili

I vincoli di interezza delle variabili nascono da esigenze diverse:

Interezza delle variabili

I vincoli di interezza delle variabili nascono da esigenze diverse:

- Necessità di rappresentare quantità indivisibili (per esempio persone, beni da produrre, ecc)

Interezza delle variabili

I vincoli di interezza delle variabili nascono da esigenze diverse:

- Necessità di rappresentare quantità indivisibili (per esempio persone, beni da produrre, ecc)
- Necessità di scegliere tra un numero finito di alternative (per esempio composizione di portafogli, assegnazione di turni lavorativi, ecc)

Interezza delle variabili

I vincoli di interezza delle variabili nascono da esigenze diverse:

- Necessità di rappresentare quantità indivisibili (per esempio persone, beni da produrre, ecc)
- Necessità di scegliere tra un numero finito di alternative (per esempio composizione di portafogli, assegnazione di turni lavorativi, ecc)
- Necessità di rappresentare vincoli logici (per esempio un certo vincolo subentra nel problema solo se una determinata variabile assume determinati valori)

Variabili intere in AMPL

In AMPL, per vincolare una variabile ad assumere valori interi è sufficiente scrivere nel file `.mod`:

```
var x >=0, integer;
```

oppure

```
var x binary;
```

se la variabile è binaria, cioè può assumere solo valore 0-1.

Variabili intere in AMPL

In AMPL, per vincolare una variabile ad assumere valori interi è sufficiente scrivere nel file `.mod`:

```
var x >=0, integer;
```

oppure

```
var x binary;
```

se la variabile è binaria, cioè può assumere solo valore 0-1.

Attenzione: utilizzare solutori per PLIM (gurobi, cplex, xpress...)

Scelta ottima di mezzi di trasporto (i)

Esempio 5.1

Un carico di merce deve essere trasportato lungo una rotta passante per 5 diverse città, avendo a disposizione 3 diversi tipi di trasporto: ferroviario, stradale e aereo. In ogni città intermedia si può effettuare un cambio. Nella seguente tabella si riportano i costi di trasporto per le varie tratte.

	1-2	2-3	3-4	4-5
Ferroviario	30	25	40	60
Stradale	25	40	45	50
Aereo	40	20	50	55

Il costo del cambio è riportato nella seguente tabella.

	Ferroviario	Stradale	Aereo
Ferroviario	0	5	12
Stradale	8	0	10
Aereo	15	10	0

Determinare il piano di trasporto che minimizza il costo totale.

Scelta ottima di mezzi di trasporto (ii)

Variabili:

- x_{ik} è una variabile binaria che indica se il mezzo di trasporto i viene utilizzato per la tratta k,;
- y_{ijk} è una variabile binaria che indica se, dalla tratta k alla tratta $k + 1$, avviene un cambio tra il mezzo di trasporto i e il mezzo di trasporto j.

Siano:

- c_trasp_{ik} il costo di trasporto per il mezzo i lungo la tratta k,
- c_cambio_{ij} il costo per il cambio tra il mezzo di trasporto i e il mezzo di trasporto j.

Scelta ottima di mezzi di trasporto (iii)

$$\min \sum_{i=1}^{3}\sum_{k=1}^{4} c\_trasp_{ik} x_{ik} + \sum_{i=1}^{3}\sum_{j=1}^{3}\sum_{k=1}^{3} c\_cambio_{ij} y_{ijk}$$

$$\sum_{i=1}^{3} x_{ik} = 1, \qquad k = 1, \dots, 4$$

$$\sum_{i=1}^{3}\sum_{j=1}^{3} y_{ijk} = 1, \qquad k = 1, \dots, 3$$

$$x_{ik} + x_{j(k+1)} \geq 2y_{ijk} \qquad i = 1, \dots, 3, \quad j = 1, \dots, 3, \quad k = 1, \dots, 3$$

$$x_{ik} \in \{0, 1\} \qquad i = 1, \dots, 3, \quad k = 1, \dots, 4$$

$$y_{ijk} \in \{0, 1\} \qquad i = 1, \dots, 3, \quad j = 1, \dots, 3, \quad k = 1, \dots, 3$$

Scelta ottima di mezzi di trasporto (iii)

$$\min \sum_{i=1}^{3}\sum_{k=1}^{4} c\_trasp_{ik} x_{ik} + \sum_{i=1}^{3}\sum_{j=1}^{3}\sum_{k=1}^{3} c\_cambio_{ij} y_{ijk}$$

$$\sum_{i=1}^{3} x_{ik} = 1, \qquad k = 1,\dots,4$$

$$\sum_{i=1}^{3}\sum_{j=1}^{3} y_{ijk} = 1, \qquad k = 1,\dots,3$$

$$x_{ik} + x_{j(k+1)} \geq 2y_{ijk} \qquad i = 1,\dots,3, \quad j = 1,\dots,3, \quad k = 1,\dots,3 \quad (*)$$

$$x_{ik} \in \{0,1\} \qquad i = 1,\dots,3, \quad k = 1,\dots,4$$

$$y_{ijk} \in \{0,1\} \qquad i = 1,\dots,3, \quad j = 1,\dots,3, \quad k = 1,\dots,3$$

$(*)$ in alternativa a

$$x_{ik} \geq y_{ijk} \qquad i = 1,\dots,3, \quad j = 1,\dots,3, \quad k = 1,\dots,3$$

$$x_{j(k+1)} \geq y_{ijk} \qquad i = 1,\dots,3, \quad j = 1,\dots,3, \quad k = 1,\dots,3$$

Scelta ottima di mezzi di trasporto (iii)

$$\min \sum_{i=1}^{3}\sum_{k=1}^{4} c\_trasp_{ik} x_{ik} + \sum_{i=1}^{3}\sum_{j=1}^{3}\sum_{k=1}^{3} c\_cambio_{ij} y_{ijk}$$

$$\sum_{i=1}^{3} x_{ik} = 1, \qquad k = 1, \ldots, 4 \quad (**)$$

$$\sum_{i=1}^{3}\sum_{j=1}^{3} y_{ijk} = 1, \qquad k = 1, \ldots, 3$$

$$x_{ik} + x_{j(k+1)} \geq 2y_{ijk} \qquad i = 1, \ldots, 3, \quad j = 1, \ldots, 3, \quad k = 1, \ldots, 3 \quad (*)$$

$$x_{ik} \in \{0, 1\} \qquad i = 1, \ldots, 3, \quad k = 1, \ldots, 4$$

$$y_{ijk} \in \{0, 1\} \qquad i = 1, \ldots, 3, \quad j = 1, \ldots, 3, \quad k = 1, \ldots, 3$$

$(*)$ in alternativa a

$$x_{ik} \geq y_{ijk} \qquad i = 1, \ldots, 3, \quad j = 1, \ldots, 3, \quad k = 1, \ldots, 3$$

$$x_{j(k+1)} \geq y_{ijk} \qquad i = 1, \ldots, 3, \quad j = 1, \ldots, 3, \quad k = 1, \ldots, 3$$

$(**)$ ridondante in presenza del secondo gruppo di vincoli (somma delle y a 1)

Scelta ottima di mezzi di trasporto (iii)

$$\min \sum_{i=1}^{3} \sum_{k=1}^{4} c\_trasp_{ik} x_{ik} + \sum_{i=1}^{3} \sum_{j=1}^{3} \sum_{k=1}^{3} c\_cambio_{ij} y_{ijk}$$

$$\sum_{i=1}^{3} x_{ik} = 1, \qquad k = 1, \ldots, 4 \quad {\color{red}(**)}$$

$$\sum_{i=1}^{3} \sum_{j=1}^{3} y_{ijk} = 1, \qquad k = 1, \ldots, 3 \quad {\color{red}(\flat)}$$

$$x_{ik} + x_{j(k+1)} \geq 2 y_{ijk} \qquad i = 1, \ldots, 3, \quad j = 1, \ldots, 3, \quad k = 1, \ldots, 3 \quad {\color{red}(*)} \quad {\color{red}(\flat)}$$

$$x_{ik} \in \{0, 1\} \qquad i = 1, \ldots, 3, \quad k = 1, \ldots, 4$$

$$y_{ijk} \in \{0, 1\} \qquad i = 1, \ldots, 3, \quad j = 1, \ldots, 3, \quad k = 1, \ldots, 3$$

$(*)$ in alternativa a $\quad x_{ik} \geq y_{ijk} \quad i = 1, \ldots, 3, \quad j = 1, \ldots, 3, \quad k = 1, \ldots, 3$

$\quad x_{j(k+1)} \geq y_{ijk} \quad i = 1, \ldots, 3, \quad j = 1, \ldots, 3, \quad k = 1, \ldots, 3$

$(**)$ ridondante in presenza del secondo gruppo di vincoli (somma delle y a 1)

$(\flat)$ in alternativa a $\quad x_{ik} + x_{j(k+1)} \leq 1 + y_{ijk} \quad i = 1, \ldots, 3, `; j = 1, \ldots, 3, \; k = 1, \ldots, 3$

Scelta ottima di mezzi di trasporto (iv)

multimodale.mod

```
param n_tratte;

set TRASP;
set TRATTE := 1..n_tratte ordered;
set SUB_TRATTE := TRATTE diff {n_tratte};

param c_trasp{TRASP,TRATTE};
param c_cambio{TRASP,TRASP};

var x{TRASP,TRATTE} binary;
var y{TRASP,TRASP,SUB_TRATTE} binary;

minimize f :
sum{i in TRASP,k in TRATTE} c_trasp[i,k]*x[i,k]
+ sum{i in TRASP, j in TRASP, k in SUB_TRATTE} c_cambio[i,j]*y[i,j,k];

s.t. v_trasp{k in TRATTE} : sum{i in TRASP} x[i,k] = 1;
s.t. v_cambi{k in SUB_TRATTE} : sum{i in TRASP, j in TRASP} y[i,j,k] = 1;
s.t. v_xy{i in TRASP, j in TRASP, k in SUB_TRATTE} :
     x[i,k] + x[j,next(k,TRATTE)] >= 2*y[i,j,k];
```

Scelta ottima di mezzi di trasporto (v)

multimodale.dat

```
param n_tratte := 4;

set TRASP := Ferroviario Stradale Aereo;

param c_trasp : 1  2  3  4 :=
Ferroviario    30 25 40 60
Stradale       25 40 45 50
Aereo          40 20 50 55;

param c_cambio : Ferroviario Stradale Aereo :=
Ferroviario              0        5      12
Stradale                 8        0      10
Aereo                   15       10       0;
```

Scelta ottima di mezzi di trasporto (vi)

Esempio 5.1 (cont.)

Verificare la soluzione del rilassamento continuo e il processo di risoluzione nei casi determinati da:

- vincoli logici originari, alternativa $(*)$, alternativa $(\flat)$
- presenza o assenza dei vincoli ridondanti $(**)$

Suggerimenti:

- Si può risolvere il rilassamento continuo [o il MILP] usando
  `option relax_integrality 1 [0]`
- Per avere più informazioni sul processo di risoluzione (con Cplex)
  `option cplex_options 'mipdisplay=X'` [con `X=0,1,`**`2`**`,3,...`]
- Per eliminare [ripristinare] i vincoli `nomevincolo{espressione indicizzante}`, usare
  `drop [restore] nomevincolo{espressione indicizzante}`

Scelta ottima di mezzi di trasporto (vii)

multimodale.run

```
reset;
model multimodale.mod;
data multimodale.dat;
printf("\n\n\n--------- MILP ------------------\n");
option solver cplex;
option cplex_options 'mipdisplay=2';
option relax_integrality 0;
solve;
display f, x, y;

printf("\n\n\n--------- LP 1 ------------------\n");
option cplex_options 'mipdisplay=0';
option relax_integrality 1;
solve;
display f, x, y;

printf("\n\n\n--------- LP 2 ------------------\n");
drop v_trasp;
option relax_integrality 1;
solve;
display f, x, y;
restore v_trasp;
```

Esercizio proposto (i)

Esercizio 5.1

Una centrale elettrica deve assicurare una produzione giornaliera di almeno 4000 megawatt di giorno e di almeno 2800 megawatt di notte. Tale energia può essere prodotta da tre diversi generatori a costi diversi, che dipendono anche dal periodo di utilizzo di ciascun generatore (giorno o notte). In particolare, nella seguente tabella sono riportati i costi di attivazione (in Euro), i costi di ogni megawatt prodotto (in Euro) e la capacità massima di ogni generatore (in megawatt), dove ogni capacità è da intendersi relativa al periodo considerato (giorno o notte).

	Costo attivazione		Costo per megawatt	Capacità max
	giorno	notte		
Generatore A	750	1000	3	2000
Generatore B	600	900	5	1700
Generatore C	800	1100	6	2500

Formulare un modello per determinare quanta energia produrre da ogni generatore di giorno e di notte in modo da minimizzare i costi totali e rispettare i vincoli. Considerare anche il caso in cui la produzione notturna avviene per multipli di 8.5 megawatt.

Esercizio proposto (ii)

Variabili:

- x_{ij} è la quantità di megawatt prodotta dal generatore i nel periodo j, $i \in \{A, B, C\}$, $j \in \{1, 2\}$;
- δ_{ij} è una variabile binaria che indica se il generatore i è usato o meno nel periodo j, $i \in \{A, B, C\}$, $j \in \{1, 2\}$;

$$
\begin{aligned}
\min\ & 3x_{A1} + 3x_{A2} + 5x_{B1} + 5x_{B2} + 6x_{C1} + 6x_{C2} + \\
& +750\delta_{A1} + 1000\delta_{A2} + 600\delta_{B1} + 900\delta_{B2} + 800\delta_{C1} + 1100\delta_{C2} \\
& x_{A1} + x_{B1} + x_{C1} \geq 4000 \\
& x_{A2} + x_{B2} + x_{C2} \geq 2800 \\
& x_{A1} \leq 2000\delta_{A1} \qquad x_{A2} \leq 2000\delta_{A2} \\
& x_{B1} \leq 1700\delta_{B1} \qquad x_{B2} \leq 1700\delta_{B2} \\
& x_{C1} \leq 2500\delta_{C1} \qquad x_{C2} \leq 2500\delta_{C2} \\
& x_{i2} = 8.5\, y_{i2}, \quad i \in \{A, B, C\} \\
& x_{Aj} \geq 0, x_{Bj} \geq 0, x_{Cj} \geq 0, \quad j = 1, 2 \\
& \delta_{Aj} \in \{0, 1\}, \delta_{Bj} \in \{0, 1\}, \delta_{Cj} \in \{0, 1\} \quad j = 1, 2 \\
& y_{i2} \in \mathbb{Z}_+, \quad i \in \{A, B, C\}
\end{aligned}
$$

Esercizio proposto (iii)

centrale.mod

```
set GENERATORI; set PERIODI;

param c_att{GENERATORI,PERIODI}>=0;
param c_mw{GENERATORI}>=0;
param cap{GENERATORI};
param prod_min{PERIODI};
param uni_prod > 0;

var x{GENERATORI,PERIODI} >=0;
var d{GENERATORI,PERIODI} binary;
var y{i in GENERATORI, j in PERIODI: j == 'notte'} integer >=0;

minimize f : sum{i in GENERATORI, j in PERIODI}
(c_mw[i]*x[i,j]+c_att[i,j]*d[i,j]);

s.t. v_prod{j in PERIODI} :
sum{i in GENERATORI} x[i,j] >= prod_min[j];

s.t. v_cap{i in GENERATORI, j in PERIODI} :
x[i,j] <= cap[i]*d[i,j];

s.t. v_uni{i in GENERATORI, j in PERIODI: j == 'notte'} :
x[i,j] = uni_prod*y[i,j];
```

Esercizio proposto (iv)

centrale.dat

```
set GENERATORI := A B C;
set PERIODI := giorno notte;

param : c_mw cap :=
A        3      2000
B        5      1700
C        6      2500;

param prod_min :=
giorno 4000
notte  2800;

param c_att : giorno notte :=
A              750   1000
B              600    900
C              800   1100;

param uni_prod = 8.5;
```

Esercizio proposto (v)

centrale.run

```
reset;
model centrale.mod;
data centrale.dat;

options solver cplex;
solve;

display f;
display x, d, y;
```